image[DORA,FUNS]

Johan G. F. Belinfante
2004 August 13

```
In[1]:= << goedel60.11c; << tools.m;

        :Package Title: goedel60.11c           2004 August 11 at 11:15 p.m.

        It is now:  2004 Aug 13 at 10:29

        Loading Simplification Rules

        TOOLS.M                       Revised 2004 August 11

        weightlimit = 40

In[2]:= Begin["Goedel‘Private‘"];
```

summary

This notebook is concerned with the relation **image[DORA,FUNS]** and its relation to the equipollence relation **Q**. Recall that an ordered pair **pair[u,v]** belongs to the relation **image[DORA,FUNS]** if there is a function with domain **u** and range **v**.

```
In[3]:= class[pair[u, v], exists[x, and[FUNCTION[x], equal[u, domain[x]], equal[v, range[x]]]]]

Out[3]= image[DORA, FUNS]
```

The equipollence relation **Q** is defined in a like manner, but with the added requirement that the function be one−to−one.

```
In[4]:= class[pair[u, v], exists[x, and[ONEONE[x], equal[u, domain[x]], equal[v, range[x]]]]]

Out[4]= Q
```

When one assumes tha axiom of choice, there is a simple connection between these two relations and one can reduce the study of **image[DORA, FUNS]** to that of **Q**. The question is whether one can make a similar reduction without using the axiom of choice. This weak result was deduced previously, but it does not provide an equation:

```
In[5]:= subclass[image[DORA, FUNS], composite[inverse[POWER], Q, inverse[S], POWER]]

Out[5]= True
```

A feeble step in the direction of obtaining an equation is taken in the present notebook, based on the fact that any function can be written as the composite of a restrtiction of **SECOND** and a bijection. This fact is conveniently derived from the following identity:

```
In[6]:= composite[SECOND, id[x], inverse[FIRST]]

Out[6]= composite[Id, x]
```

As a corollary it is shown that the relation **image[DORA,FUNS]** can be factored into **image[DORA, P[SECOND]]** and the equipollence relation **Q**. Of course this reduction is only useful if one can derive interesting properties of the relation o **image[DORA, P[SECOND]]**.

review of some old facts

The **GOEDEL** program already contains a limited amount of information about the relation **image[DORA, FUNS]**:

```
In[7]:= idempotent[image[DORA, FUNS]]

Out[7]= True

In[8]:= commute[S, image[DORA, FUNS]]

Out[8]= True

In[9]:= composite[S, image[DORA, FUNS]]

Out[9]= union[cart[V, complement[singleton[0]]], cart[singleton[0], V]]
```

The relation **Q** also commutes with **image[DORA, FUNS]**.

```
In[10]:= composite[Q, image[DORA, FUNS]]

Out[10]= image[DORA, FUNS]
```

From Cantor’s theorem one deduces:

```
In[11]:= subclass[image[DORA, FUNS], composite[complement[S], POWER]]

Out[11]= True
```

This fact can also be restated as an equation as follows:

```
In[12]:= SubstTest[disjoint, image[DORA, FUNS],
          complement[x], x -> composite[complement[S], POWER]]

Out[12]= equal[0, intersection[composite[S, POWER], image[DORA, FUNS]]] == True

In[13]:= intersection[composite[S, POWER], image[DORA, FUNS]] := 0
```

derivation of the stated factorization

When **x** is a function, **composite[id[x], inverse[FIRST]]** is a bjection which maps each point of the domain to the corresponding point on the graph of the function.

```
In[14]:= ONEONE[composite[id[x], inverse[FIRST]]]

Out[14]= FUNCTION[composite[Id, x]]
```

The following result can be cleaned up:

```
In[15]:= SubstTest[implies, member[x, y],
          subclass[image[z, singleton[x]], image[z, y]], z → DORA]

Out[15]= or[member[pair[domain[x], range[x]], image[DORA, y]],
           not[member[x, y]], not[member[domain[x], V]], not[member[range[x], V]]] == True

In[16]:= (% /. {x → x_, y → y_}) /. Equal → SetDelayed
```

Two unneeded hypotheses can be removed from this:

```
In[17]:= Map[not, SubstTest[and, implies[p1, p2],
           implies[p1, p3], implies[and[p1, p2, p3], p4], not[implies[p1, p4]],
           {p1 → member[x, y], p2 → member[domain[x], V], p3 → member[range[x], V],
            p4 -> member[pair[domain[x], range[x]], image[DORA, y]]}]]

Out[17]= or[member[pair[domain[x], range[x]], image[DORA, y]], not[member[x, y]]] == True

In[18]:= or[member[pair[domain[x_], range[x_]], image[DORA, y_]], not[member[x_, y_]]] := True
```

This is applied to the bjiection between domain and graph of a function.

```
In[19]:= SubstTest[implies, member[u, v], member[pair[domain[u], range[u]], image[DORA, v]],
          {u -> composite[id[funpart[x]], inverse[FIRST]], v -> BIJ}]

Out[19]= or[member[pair[domain[funpart[x]], funpart[x]], Q], not[member[funpart[x], V]]] == True

In[20]:= or[member[pair[domain[funpart[x_]], funpart[x_]], Q],
           not[member[funpart[x_], V]]] := True
```

The result is also applied to the restriction of **SECOND** that appears in the factorization.

```
In[21]:= SubstTest[implies, member[u, v], member[pair[domain[u], range[u]], image[DORA, v]],
          {u -> composite[SECOND, id[funpart[x]]], v -> P[SECOND]}]

Out[21]= or[member[pair[funpart[x], range[funpart[x]]], image[DORA, P[SECOND]]],
           not[member[funpart[x], V]]] == True

In[22]:= or[member[pair[funpart[x_], range[funpart[x_]]], image[DORA, P[SECOND]]],
           not[member[funpart[x_], V]]] := True

In[23]:= SubstTest[implies, and[member[pair[u, v], composite[Id, t]],
           member[pair[v, w], composite[Id, s]]], member[pair[u, w], composite[s, t]],
          {u → domain[funpart[x]], v → funpart[x], w → range[funpart[x]],
           s → image[DORA, P[SECOND]], t → Q}]

Out[23]= or[member[pair[domain[funpart[x]], range[funpart[x]]],
            composite[image[DORA, P[SECOND]], Q]], not[member[funpart[x], V]],
           not[member[pair[domain[funpart[x]], funpart[x]], Q]],
           not[member[pair[funpart[x], range[funpart[x]]], image[DORA, P[SECOND]]]]] == True

In[24]:= (% /. x → x_) /. Equal → SetDelayed

In[25]:= Map[not, SubstTest[and, implies[p1, p2],
           implies[p1, p3], implies[and[p1, p2, p3], p4], not[implies[p1, p4]],
           {p1 → member[funpart[x], V], p2 -> member[pair[domain[funpart[x]], funpart[x]], Q],
            p3 -> member[pair[funpart[x], range[funpart[x]]], image[DORA, P[SECOND]]],
            p4 -> member[pair[domain[funpart[x]], range[funpart[x]]],
              composite[image[DORA, P[SECOND]], Q]]}]]

Out[25]= or[member[pair[domain[funpart[x]], range[funpart[x]]],
            composite[image[DORA, P[SECOND]], Q]], not[member[funpart[x], V]]] == True

In[26]:= (% /. x → x_) /. Equal → SetDelayed
```

The **funpart** wrapper is removed:

```
In[27]:= SubstTest[implies, equal[y, funpart[x]],
          or[member[pair[domain[y], range[y]], composite[image[DORA, P[SECOND]], Q]],
           not[member[y, V]]], y → x]

Out[27]= or[member[pair[domain[x], range[x]], composite[image[DORA, P[SECOND]], Q]],
           not[FUNCTION[x]], not[member[x, V]]] == True

In[28]:= (% /. x → x_) /. Equal → SetDelayed

In[29]:= Map[equal[V, #] &, SubstTest[class, x, implies[member[x, u], member[x, v]], {u → FUNS,
             v -> image[inverse[DORA], composite[image[DORA, P[SECOND]], Q]]}]] // Reverse

Out[29]= subclass[image[DORA, FUNS], composite[image[DORA, P[SECOND]], Q]] == True

In[30]:= % /. Equal → SetDelayed
```

The reverse inclusion follows from the idempotence of the relation **image[DORA,FUNS]**:

```
In[31]:= SubstTest[implies, and[subclass[u, w], subclass[v, w]],
          subclass[composite[u, v], composite[w, w]],
          {u -> image[DORA, P[SECOND]], v -> Q, w → image[DORA, FUNS]}]

Out[31]= subclass[composite[image[DORA, P[SECOND]], Q], image[DORA, FUNS]] == True

In[32]:= % /. Equal → SetDelayed
```

Combining these two inclusions yields an equation that can be added as a new rewrite rule:

```
In[33]:= SubstTest[and, subclass[u, v], subclass[v, u],
          {u -> composite[image[DORA, P[SECOND]], Q], v -> image[DORA, FUNS]}]

Out[33]= True == equal[composite[image[DORA, P[SECOND]], Q], image[DORA, FUNS]]

In[34]:= composite[image[DORA, P[SECOND]], Q] := image[DORA, FUNS]
```

reference

The following reference can be consulted for a much deeper and more detailed discussion about what can be done without the axiom of choice.

```
In[36]:= "Lorenz Halbeisen and Saharon Shelah, Relations
            between some cardinals in the absence of the Axiom of Choice,
            Bulletin of Symbolic Logic, vol. 7 (June 2001), pages 237-261";
```